
Introduction
LLC resonant half bridge converters are widely used in

consumer electronics, like powering the display panel of
LCD TV. However, the operating principles of a LLC
resonant half bridge converter are far from apparent and
intuitive [1]. This application note focuses on the intuitive
and conceptual understanding of a LLC resonant converter.
Some basic mathematical equations are also shown in this
application note for fundamental quantitative understanding
of LLC resonant half bridge converters.

Several key topics, including evolution of LLC resonant
half bridge converter, basic analyzing method, DC
characteristics, features of typical switching waveforms,
soft−switching feature, and CCM/DCM features are
covered in this application note.

For hands−on understanding of the LLC resonant half
bridge converter evaluation board provided by
ON Semiconductor, spice simulation circuits are provided
for transient simulation and DC simulation.

Evolution of LLC Resonant Half bridge Converter
Figure 1 is a typical LLC resonant converter. Values of

major circuit components are also shown in it.

Figure 1. Transient Simulation Circuit of LLC
Resonant Half Bridge Converter

While the LLC resonant half bridge converter is
complicated, its intrinsic structure is very simple. It starts
from a voltage divider and then adds isolation and
rectification to output. With the half bridge input, the LLC
resonant half bridge converter topology is formed.

The evolution has the following procedures as shown in
the next three sections.

Regular Voltage Divider and Voltage Divider with
Resonant Tank

Figure 2. A Voltage Divider

Figure 3. An AC Voltage Divider Formed by a
Resonant Tank

As shown in Figure 2, a regular voltage divider converts a
higher voltage to a lower voltage. The same effect can be
achieved by a resonant tank and a load resistance [2], as
shown in Figure 3. In Figure 3, the steady state output voltage
is about the same as the input voltage when the circuit
operates at its resonant frequency. The output voltage is less
than input voltage when circuit is not operating at its resonant
frequency. The voltage divider ratio is frequency dependent.

Square Wave Input and Transformer Isolated Output

Figure 4. A DC/AC Resonant Converter

Figure 5. A DC/AC Resonant Converter with
Transformer Isolated Load

Now as shown in Figure 4, the input of a voltage divider
is formed by square wave, rather than a single frequency
sinusoidal input. At steady state, the high frequency
harmonic waveforms of square wave input are filtered by the
resonant tank, which results in sinusoidal load current. The
sinusoidal output current is then isolated from resonant tank
by a transformer, which results in circuit shown in Figure 5.

Output Rectification and Soft Switching at Input

Figure 6. A DC/DC Resonant Converter

Figure 7. A DC/DC Resonant Converter with Soft
ZVS Switching

To obtain DC voltage, output current of Figure 5 is
rectified and stabilized by a large capacitance, providing a
stable DC voltage and power on the load, as shown in
Figure 6. When the circuit in Figure 6 operates at resonant
frequency, the input current and output current are exactly
in phase, which means that both high side and low side
switches are switching with zero current. However, this
benefit does not exist when the switching frequency is
deviates from resonant frequency. To achieve soft switching
in the vicinity of the resonant frequency, a magnetizing
current is developed within the transformer. This
magnetizing current the phase node to ground before the low
side turns on and raises the phase node to the input voltage
before the high side turns on, creating ZVS while switching
on. By controlling the magnetizing current of transformer,
the switching off loss of high side and low side can be
minimized.

As a voltage divider, the output voltage is less than the
input voltage. However, a LLC resonant converter can
operates in a mode that increases rather than decreases the
voltage. On the condition that Lm participates in the
resonant tank as a resonant inductor, the output voltage is the
voltage of resonant inductor, which could results in output
voltage much higher than the input voltage.

A regular LLC resonant half bridge converter works a
combination mode of a voltage divider and amplifier of
resonant inductor voltage of the resonant tank.

At the resonant frequency, impedance of resonant tank is
zero, which means the input voltage is 100% applied on the
load. As frequency deviates from the resonant frequency, the
impedance of resonant tank is larger and larger, which
means the voltage on load is lower and lower. By varying the
operating frequency, the output power can be controlled.

Some basic features of LLC resonant half bridge
converter include HS&LS are 50% duty cycle, frequency
variation controls load current, CCM/DCM is defined by
rectifier current, not inductor current, and magnetizing
current of transformer is the soft switching current for ZVS
switching of Q1 and Q2.

Basic Analysis Method

Fundamental Harmonics Approximation [3]
In application, LLC resonant converter works around the

frequency of resonant tank (formed by Lr, Cr). Due to the
nature of LC filter, only the fundamental frequency of the
input voltage can pass through the filter. All the harmonics
are filtered by the resonant tank. Based on this fact, the
topology of LLC resonant converter could be simplified to
the circuit shown in Figure 8. In Figure 8, “Rac” is the
reflected load from secondary side of transformer to primary
side of transformer when only the fundamental harmonics is
considered.

Rac � 8
�2 � N2 � RLOAD (eq. 1)

Here, N is the transformer turn ratio of primary side and
secondary side.

Two Operating Modes [4]
Depending on whether transformer is driving power to the

output or not, the circuit of Figure 8 has two operating
modes. One mode is Rac=0, which means the magnetizing
inductance is cut off from circuit. This interval happens
when the transformer is driving power to the output,
resulting in the primary side of the transformer been
clamped to a constant voltage (the output voltage reflected
to primary side of transformer). The other mode is that
Rac=8, which means Rac is cut off from circuit. This mode
happens when transformer stops driving power to output.
Both modes are shown in Figure 9 and Figure 10 respectively.

Figure 8. Simplified LLC Resonant Converter
under Fundamental Harmonics Appropriation

Figure 9. Operating
Mode #1: Rac = 0

Figure 10. Operating
Mode #2: Rac = �

It is interesting to see that, for the operating mode #1, the
maximum output voltage is equal to input voltage, which
happens when impedance of resonant tank is zero. For the
operating mode #2, the output voltage is the voltage on
magnetizing inductor, which is the resonant voltage of the
magnetizing inductance and could be way higher than input
voltage. This is the basic DC characteristics of LLC resonant
converter.

In regular operating mode of LLC resonant converter,
0<=Rload<∞, both operating modes happen in one
switching cycle.

Two Resonant Frequency of LLC Resonant Converters
[5]

There are two resonant frequencies in a LLC resonant half
bridge converter. One is resonant frequency at operating
mode #1, shown in Figure 9 operating mode #1: Rac=0. The
resonant frequency of mode #1 is expressed by Equation 2.
In the vicinity of fr1, output voltage is equal or less than
input voltage.

fr1 � 1
2� Lr � Cr� (eq. 2)

The other frequency is the resonant frequency at operating
mode #2, shown in Figure 10. The resonant frequency of
mode #2 is expressed by Equation 3. In the vicinity of fr2,
the output voltage is higher than input voltage.

fr2 � 1
2� (Lr � Lm) � Cr�

(eq. 3)

Two Key Parameters: m and Q
The operating region of LLC resonant half bridge

converters is in the vicinity of fr1. In order to quantitatively
understand the impact of gain at frequency fr2 on gain at fr1,
a parameter, m, is introduced to describe the distance
between two frequencies fr1 and fr2. Equation 4 is the
definition of “m”.

m � Lm
Lr

(eq. 4)

When m=0, it means Lm=0 and the distance from fr1 and
fr2 is zero. However, at this condition the voltage gain at
frequency fr2 is also zero. Therefore, the impact of the gain
at fr2 on the gain at fr1 is zero. At this condition, the
maximum gain is always 1 and happens at resonant
frequency fr1.

When m=0, it means Lm=0, the distance from fr1 to fr2
is fr1. Gain at fr2 has a certain impact on gain at fr1.
Assuming the frequency fr1 is high enough to be out of the
selection band of the resonant tank (formed by Lr+Lm and

Cr) at resonant frequency fr2, the impact of the gain at fr2 on
the gain at fr1 is nearly zero as well. This assumption is
generally true because fr1 is usually around 200 kHz, which
is more than 5 decades away from 1 Hz.

For the two most extreme cases shown above, the impact
of the gain at fr2 on the gain at fr1 is both zero. The most
impact of the gain at fr2 on the gain at fr1 comes when
0<m<∞. This feature is shown in Figure 11. This impact is
elaborated in the next section.

In a LC resonant circuit, quality factor Q is the parameter
to describe how fast the resonant gain drops when deviating
from the resonant frequency. The definition of Q is shown
by Equation 5.

Q �

Lr
Cr

�
Rac

(eq. 5)

Here, Rac is the load resistance reflected to primary side
of transformer under the fundamental harmonic
approximation, shown in Equation 1.

At a specific fr1, higher Q means faster voltage gain drop
when deviating from fr1. The higher the Q, the less the gain
at fr2 impacts the gain at fr1. Q directly impacts the voltage
gain between fr1 and fr2.

In a LLC circuit design, Q needs to be controlled within
a certain range, so that the voltage stress on the resonant
inductance or resonant capacitor is under minimized.

DC Characteristics and Operation Point
Placement

To understand the operation of LLC resonant circuits, it is
necessary to understand DC characteristics at various
frequencies, especially in the vicinity of fr1 and fr2.

A frequency sweep for the circuit shown in Figure 8 yields
the DC characteristics of the resonant converter, as shown in
Figure 12. By varying the value of the load resistance, a
family of DC gain curves over a frequency range is created.

Figure 11. Gain at fr1 and Impact Ratio of Gain at
fr1 on Gain at fr2 as a Function of the Distance

between fr1 and fr2

The most apparent feature of the DC characteristics
shown in Figure 12 is the two resonant frequencies, fr2 and
fr1. These two resonant frequencies divide the frequency
range into three regions. Given that a resonant tank exhibits
capacitive characteristics when below its resonant
frequency and inductive characteristics when above, three
regions can be identified in Figure 13: capacitive for fr1 and
capacitive for fr1 and inductive for fr2, and inductive for
both fr1 and fr2.

 http://www.onsemi.com/


AND9408/D

www.onsemi.com
4

Figure 12. DC Characteristics of LLC Resonant
Half Bridge Converter

Figure 13. Three Regions of DC Characteristics of
LLC Resonant Half Bridge Converter

In the region between fr1 and fr2, there is a competition
between capacitive characteristics of fr1 and inductive
characteristics of fr2. This competition results in the top half
of the region is inductive, in which gain decreases as
frequency increases, the bottom half is capacitive, in which
gain increases as frequency increases. This result is shown
in Figure 12, where section #1 is marked as capacitive
region, section #2 is marked as inductive region, and section
#3 is marked as inductive region.

The regular operating point of LLC resonant converter
resides in section #2 near fr1 due to lower switching
frequency and stable voltage gain. Here, one can see that all
the curves go through the point of (fr1, 1), which means that
the operating frequency won’t change too much while load
varies. This feature limits operating frequency into a relative
narrow range and gives better stability.

Various Operating Mode of LLC Resonant Half
Bridge Converter

There are many operating modes of LLC resonant
converters. The most frequently used operating mode is in
the vicinity of fr1. It is of greater interest to analyze the
operation exactly at frequency fr1 and slightly higher and
lower than fr1.

Operation Mode #1: fs=fr1
This mode is the basis of all other modes. At this mode,

ringing current of resonant tank is exactly in phase with load
current.

Resonant Cycle and Operation Intervals
1. high side turn−off to low side turn−on

When the high side is turning off, the primary current of
the transformer (blue line) resonates back to magnetizing
current (coral), as shown in Figure 14. The resonant current
flows positively from phase node to ground. After the high
side is fully turned off, the resonant current freewheels
through diode of low side MOSET, which clamps
drain−source voltage of low side MOSFET to zero.

As the resonance current drops below the magnetizing
current, the primary current of the transformer become
negative, driving current through D2 to load (assuming
magnetizing current is sufficiently small). From this
moment, the secondary side of transformer starts rectifying
the negative waveforms. D2 starts to conduct load current,
while D1 is off.

Figure 14. Waveforms of LLC Resonant Half
Bridge Converter

Blue: resonant current, red: driving pulse of high side MOSFET, 
Green: driving pulse of low side MOSFET, 
Magenta: current of D1, 
Cyan: current of D2

2. low side turn−on to low side turn−off
After the drain−source voltage of the low side MOSFET

is clamped to zero volts, the low side MOSFET is turned−on.
This switching−on transient is Zero Voltage switching.
During this interval, the resonant current starts decreasing
from magnetizing current reaching negative current.
Transformer current on primary side reverses direction and
stays negative for 180 degrees.

3. low side turn−on to high side turn−on
At the moment of low side turn−off, the resonant current

is negative and equal to magnetizing current in magnitude.
The negative current flows through body diode of the high
side MOSSFET clamping high side MOSFET’s
drain−source voltage to zero volts. After that, the high side
MOFET is turned−on under ZVS conditions.

Operation Mode #2: fs>fr1
When Compared to fs=fr1, the waveforms of fs>fr1 are a

little different. The major difference is that at heavy load, the
resonant current is of higher magnitude than the
magnetizing current when the high side or low side is
turning−off. With this condition, when phase node switches
high or low, the applied phase node voltage will force
resonant current to go down to the level of magnetizing
current. At the same time, the voltage on primary side of
transformer remains N*Vout. The key feature at this
moment is that the resonant current drops in a straight line
with constant slope (This slope is equal to
(Vin–N*Vout)/Lr), as seen in Figures 15 and 16.

Figure 15. Rectifier Current of D1 at fs>fr1

Figure 16. Resonant Current and Magnetizing
Current on Primary Side

Figure 17 shows the sequence of phase node voltage (blue
line on top) and voltage on the primary side of MOSFETs
(oral, green, purple lines). There is a delay in phase between
phase node voltage and voltage on the primary side of
MOSFET. This delay is the interval driving resonant current
back to magnetizing current. At the end of this delay, current
of D1 and D2 alternates. As can be seen in Figure 17, the
delay is related to the load current. The higher the load
current is, the longer the delay is. This is because it takes
more time to drive the resonant current to magnetic current
at higher loads.

Figure 17. Phase Node Voltage and Primary Side
Voltage of Transformer

Operation Mode #3: fs<fr1
Unlike the situation fs>fr1, in which resonant current has

not fallen down to the magnetizing current when the
MOSFET are switching, the resonant current has already
fallen down to the magnetizing current before the MOSFET
is switching, as is shown in Figure 18. After that, if the
resonant current is high enough, two rectifier diodes conduct
current alternatively, which is the CCM. If resonant current
is not high enough to turn−on the rectifier, the rectifier will
go into DCM mode. This is the key characteristic of
switching below fr1. CCM/DCM is elaborated in a later
section.

Figure 18. Current of D1 at fs<fr1

Figure 19. Resonant Current and Magnetizing
Current

Figure 19 shows resonant current and magnetizing current
on the primary side of the transformer. As can be seen, the
resonant current resonates back to magnetizing current
before the high side is turned−off, leaving an interval for
DCM mode or CCM mode, depending on the resonant
current level. This interval is more clearly seen in Figure 20.
The voltage of primary side (oral, light blue and purple lines)
drops to zero or an intermediate voltage after the resonant
current drops to the magnetizing current. After that, if
resonant current is high, the magnetizing inductor on
primary side of transformer is either functioning sinusoidal
with resonant tank or clamped by the output voltage if
resonant current is high enough to drive current to the load.

Figure 20. Phase Node Voltage and Primary Side
Voltage of Transformer

Zero Voltage−Switching and Zero Current
Switching Features

The soft switching turn−on is achieved by the magnetizing
current of transformer, which makes the circuit working in
inductive condition, as shown in Figure 21. For the
capacitor, ZCS turn−off could be achieved, as shown in
Figure 22.

The transformer is designed to have sufficient magnetizing
inductance to maintain the magnetizing current. The smaller
magnetizing inductance is, the higher magnetizing current
is. This behavior of magnetizing inductance is exactly the
same with regular inductance. Without magnetizing current,
the switching of the high side and low side MOSFETs is zero
current switch−on at fr1. If the switching frequency is not
exactly at resonant frequency fr1, soft switching−on cannot
be achieved for the low or zero magnetizing current cases,
as shown in Figure 23 and Figure 24.

Figure 21. Phase
Relation of Voltage and
Current of an Inductor

Figure 22. Phase
Relation of Voltage and
Current of a Capacitor

V, I

Inductive switching , current lag behind voltage

ZVS switching −on interval
V, I

Capacitive switching , current is ahead of voltage

ZCS switching −off interval

For the rectifier on the second side of the transformer it is
always zero current switching, which means Qrr loss is zero.

Figure 23. Switching
with Sufficiency

Magnetizing Current

Figure 24. Switching
without Sufficiency
Magnetizing Current

Continuous Current Mode (CCM) and
Discontinuous Current Mode (DCM)

The CCM and DCM mode of LLC resonant converter is
not determined by inductor current, but by rectifier current.
If the rectifier is not continuously conducting current to
output, it is DCM. Otherwise, it is CCM.

The rectifier current (diode current of rectifier) is
expressed by Equation 6. As can be seen in Equation 6, ID
is positive and therefore CCM when Ir>Im. Otherwise,
DCM occurs. In another word, if the resonant current is
sufficiently high to provide current higher than magnetic
current when primary side voltage of transformer is clamped
reversely by n*Vout, it is CCM. If the resonant current is not
high enough to be higher than magnetic current when
voltage is clamped reversely, it is DCM. If magnetizing
current is zero, there is no DCM.

ID � (Ir � Im) � N (eq. 6)

Here, Ir is the resonant current of Lr, Im is the magnetizing
current of the transformer.

There are many factors associated with the DCM [2], like
switching frequency, characteristics of resonant tank, input
to output ratio and load current. As a LLC resonant
converter, the major factor that relates with operating in
DCM or CCM is load current. At high load current, it tends
to work in CCM; at light load situation, it tends to work in
DCM.

In order to show the impact of load current on DCM and
CCM, three load conditions are simulated, as shown in
Figure 25. At high load currents, D1 and D2 alternately
conduct load current. This is CCM. At light load, there are
some intervals that the rectifiers do not provide any current
to the load, which is the DCM.

The simplified circuits of LLC resonant converters at
CCM/DCM are shown in Figures 9 and 10 respectively. For
LLC resonant converters working in DCM, the circuit
switches between circuits of Figures 9 and 10 in one
switching cycle.

Quantitatively, the boundary of DCM and CCM is a
function of four parameters: Vout, Vs/Vp, Iload, and Lm.

Figure 25. Current of D1 at Various Load Conditions

DC/Transient Simulation Circuits of LLC
Resonant Half Bridge Converter

Basic simulation information is shown below. These
simulation parameters are designed based on the evaluation
board NCP1397GANGEVB [6]. Here is the simulation circuit.

1. Vin = 400 V, Vout = 12 V, Iout = 0−20 A
2. Simulation circuit: Q1&Q2 are V−swtich: Ron =

200 m�, Coss = 200 pF.
3. Resonant tank is: Lr = 29 �H, Cr = 22 nF, (fr1 =

~200 kHz)
4. Transformer is: 16:1, Lm = 411 �H.
5. Output capacitance: Cout = 200 �F

Transient Simulation Circuit
The critical node names of transient simulation circuit are shown in Figure 1. This node names are used in simulation code.

Transient Simulation Code
**Simulation circuit of half bridge LLC resonant converter
**ON semiconductor

*converter parameters
.param Ts=6.667u;6.667us; 150kHz,
.param Vin=400 Iout=20 Lm=411u

*driver parameters
.param delay_hs={20n} tr_hs=20n tf_hs=20n
duty_cycle={(Ts−deadtime*2−(tr_hs+tf_hs+tr_ls+tf_ls))/2}
.param deadtime=200ns
.param delay_ls={duty_cycle+tr_hs+tf_hs+deadtime+delay_hs}
.param tr_ls=20n tf_ls=20n duty_prime={duty_cycle};

*sweeping parameters
.param RDSon={200} Cgs={760p} Cgd={5p} Cds={200p }

*Sweep parameters
.step param Iout list 5,10,15,20,

*SWITCH DRIVER
VCTRL1 10 sw PULSE(0V 12V {delay_hs} {tr_hs} {tf_hs} {duty_cycle} {Ts})
Rgh 10 hg 5

VCTRL2 20 gnd PULSE(0V 12V {delay_ls} {tr_ls} {tf_ls} {duty_prime} {Ts})
Rgl 20 lg 5;

* INPUT VOLTAGE
Vin Vin gnd DC {Vin}

*high side MOSFET
Rdh Vin hd 0.1m; used for channel current sensing
Shs hd hs hg hs Smod
.MODEL Smod VSWITCH(Ron={RDSon} Roff=100MEG Von=12V Voff=0V)
Cgshs hg hs {Cgs}
Cgdhs hg Vin {Cgd}
Cdshs Vin hs {Cds}
Dfwhs hs hd Dmod
.MODEL Dmod D (IS=1e−12, CJO=1P)
Lsh hs sw 15n

*low side MOSFET
Rdl sw ld 0.1m ; used for channel current sensing
Sls ld ls lg ls Smod
Cgsls lg ls {Cgs}
Cgdls lg sw {Cgd}
Cdsls sw ls {Cds}

Dfwls ls ld Dmod
Lsl ls gnd 15n

*here is resonant tank
Rlr sw int1 1m
Lr int1 RT 29u ic=0
Cr RT Vp 22n ic={Vin/2}

*transformer
Lp Vp int2 {Lm} ic=0
Rlp int2 gnd 1m
Ls1 Vs1 0 {Lm/256} ic=0 ; voltage ratio=sqrt(256)=16;
Ls2 0 Vs2 {Lm/256} ic=0
K1 Lp Ls1 Ls2 1

*here is the sensed magnetizing current of transformer
Elm 100 0 value={V(Vp)−V(int2)}
Lm 100 101 {Lm} ic=0
Rlm 101 0 1m

*here is output rectifier
D1 Vs1 Vout Dmod
Rd1 Vs1 0 1meg
D2 Vs2 Vout Dmod
Rd2 Vs2 0 1meg
Cout Vout 0 200u ic=12
*load
Rload Vout 0 {12/Iout}

*connect grounds at one point
Rgnd gnd 0 1u

* ANALYSIS
.TRAN 1n 1000u 993u 10n; 1.5 periods simulation
.PROBE ;−s(W(Rload))/s(W(Vin)+W(VCTRL1)+W(VCTRL2))
.END

DC Characteristics Simulation Circuit
The simulation circuit is shown in Figure 8.

**Simulation circuit of half bridge LLC resonant converter
**ON semiconductor

*converter parameters
.param variable=1
.param Lm=411u

*Sweep parameters
.step param variable list 10, 20, 40;

Vs sw 0 AC 1 0

*here is resonant tank
Rlr sw int1 1m
Lr int1 RT 29u
Cr RT Vp 22n

*transformer
Rac Vp 0 {variable}
Lp Vp int2 {Lm} ic=0
Rlp int2 0 1m

* ANALYSIS
.AC DEC 2000 1kHz 500kHz;
.PROBE
.END

 http://www.onsemi.com/


AND9408/D

www.onsemi.com
9

REFERENCES
1. Bo Yang, Topology Investigation for Front End

DC/DC Power Conversion for Distributed Power
System, 2003

2. AN2644, An introduction to resonant half bridge
converter

3. AN−6104, LLC resonant converter design using
FAN7688

4. AND8311/D, Understanding the LLC structure in
resonant applications

5. AND8255/D, A simple DC spice model for the
LLC converter

6. Datasheet of NCP1397A/B, high performance
resonant mode controller with integrated
high−voltage driver

ON Semiconductor and the         are registered trademarks of Semiconductor Components Industries, LLC (SCILLC) or its subsidiaries in the United States and/or other countries.
SCILLC owns the rights to a number of patents, trademarks, copyrights, trade secrets, and other intellectual property. A listing of SCILLC’s product/patent coverage may be accessed
at www.onsemi.com/site/pdf/Patent−Marking.pdf.  SCILLC reserves the right to make changes without further notice to any products herein.  SCILLC makes no warranty, representation
or guarantee regarding the suitability of its products for any particular purpose, nor does SCILLC assume any liability arising out of the application or use of any product or circuit, and
specifically disclaims any and all liability, including without limitation special, consequential or incidental damages.  “Typical” parameters which may be provided in SCILLC data sheets
and/or specifications can and do vary in different applications and actual performance may vary over time.  All operating parameters, including “Typicals” must be validated for each
customer application by customer’s technical experts.  SCILLC does not convey any license under its patent rights nor the rights of others.  SCILLC products are not designed, intended,
or authorized for use as components in systems intended for surgical implant into the body, or other applications intended to support or sustain life, or for any other application in which
the failure of the SCILLC product could create a situation where personal injury or death may occur.  Should Buyer purchase or use SCILLC products for any such unintended or
unauthorized application, Buyer shall indemnify and hold SCILLC and its officers, employees, subsidiaries, affiliates, and distributors harmless against all claims, costs, damages, and
expenses, and reasonable attorney fees arising out of, directly or indirectly, any claim of personal injury or death associated with such unintended or unauthorized use, even if such claim
alleges that SCILLC was negligent regarding the design or manufacture of the part.  SCILLC is an Equal Opportunity/Affirmative Action Employer.  This literature is subject to all applicable
copyright laws and is not for resale in any manner.

PUBLICATION ORDERING INFORMATION
N. American Technical Support: 800−282−9855 Toll Free
USA/Canada

Europe, Middle East and Africa Technical Support:
Phone: 421 33 790 2910

Japan Customer Focus Center
Phone: 81−3−5817−1050

AND9408/D

LITERATURE FULFILLMENT:
Literature Distribution Center for ON Semiconductor
19521 E. 32nd Pkwy, Aurora, Colorado 80011 USA
Phone: 303−675−2175 or 800−344−3860 Toll Free USA/Canada
Fax: 303−675−2176 or 800−344−3867 Toll Free USA/Canada
Email: orderlit@onsemi.com

ON Semiconductor Website: www.onsemi.com

Order Literature: http://www.onsemi.com/orderlit

For additional information, please contact your local
Sales Representative

 http://www.onsemi.com/
http://www.onsemi.com/site/pdf/Patent-Marking.pdf

